Stata 19

Psychometric meta-analysis

Meghan K. Cain, PhD
mcain@stata.com

June 19, 2026

Outline

- Meta-analysis
- Meta-analysis of correlations
- Psychometric meta-analysis
  - Small-study bias
  - Artificial dichotomization
  - Measurement error
  - Range restriction
- Two examples

Meta-analysis

- The goal of meta-analysis is to pool effect sizes across studies to summarize the overall evidence regarding a phenomenon of interest.



Meta-analysis of correlations

```
. webuse adherence, clear
. meta esize rho n, correlation
. meta forestplot
. meta galbraith
. meta summarize, subgroup(design)
. meta regress i.(design quality) meanage
. meta funnelplot
. meta trimfill, estimator(linear)
```

Psychometric meta-analysis

- Psychometric meta-analysis corrects correlations and their associated standard errors for
  - Small-study bias
  - Artificial dichotomization
  - Measurement error
  - Range restriction
- By eliminating the distorting effects of these statistical artifacts, we can obtain more accurate estimates of the mean correlation and the between-study variance (Schmidt and Hunter, 2015).

Introducing `meta psycorr`

- The `meta psycorr` command was added as part of the StataNow 19 release on January 28, 2026.

```
. meta psycorr rho n, options
```

- It is completely integrated into Stata's `meta` suite, including support for plots and subgroup analysis.

Small-study bias

- Sample correlation typically exhibits a small negative bias, which is practically negligible for studies with sample sizes of 20 or more (Schmidt, Le, and Oh 2019, 320).
- The attenuation factor of small-study bias and the corresponding corrected correlation are given by

$$a_{bias} = \frac{2n-2}{2n-1} \qquad \rho_{corrected} = \frac{\rho}{a_{bias}}$$

```
. meta psycor rho n, small
```

Small-study bias



A simple dataset

```
. clear

. set obs 1000
Number of observations (_N) was 0, now 1,000.

. matrix input R = (1,.5,.2\.5,1,.3\.2,.3,1)

. drawnorm x y z, corr(R) seed(12345)

. correlate x y z
(obs=1,000)

             |        x        y        z
-------------+---------------------------
           x |   1.0000
           y |   0.4938   1.0000
           z |   0.2422   0.3132   1.0000
```

Artificial dichotomization

- Dichotomization of a continuous variable results in a correlation that is lower than the correlation originally observed when both variables were treated as continuous.
- Example: You would like to know the correlation between score on a skills test and job performance, but job performance was only measured in two categories: satisfactory and unsatisfactory.

Artificial dichotomization

- Dichotomization of a continuous variable results in a correlation that is lower than the correlation originally observed when both variables were treated as continuous.
- Example: You would like to know the correlation between score on a skills test and job performance, but job performance was only measured in two categories: satisfactory and unsatisfactory.
  - In addition, some studies may report *t* tests rather than correlation coefficients. We can convert *t* statistics to correlation coefficients by

$$r = \frac{t}{\sqrt{t^2 + df}}$$

Artificial dichotomization

```
. egen x_dich = cut(x), group(2)

. correlate x* y
(obs=1,000)

             |        x   x_dich        y
-------------+---------------------------
           x |   1.0000
      x_dich |   0.7889   1.0000
           y |   0.4938   0.3852   1.0000
```

Artificial dichotomization

```
. egen x_dich = cut(x), group(2)

. correlate x* y
(obs=1,000)

             |        x   x_dich        y
-------------+---------------------------
           x |   1.0000
      x_dich |   0.7889   1.0000
           y |   0.4938   0.3852   1.0000

. summarize x_dich

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
      x_dich |      1,000          .5    .5002502          0          1
```

Artificial dichotomization



Artificial dichotomization



Artificial dichotomization



Artificial dichotomization

- When a proportion *p* of the observations is assigned to one of the two levels of the binary variable, the attenuation factor of dichotomization and corresponding corrected correlation are given by

$$a_{dich} = \frac{\phi\{\Phi^{-1}(p)\}}{\sqrt{p(1-p)}} \qquad \rho_{corrected} = \frac{\rho}{a_{dich}}$$

```
. meta psycor rho n, xdich() ydich()
```

Artificial dichotomization



Measurement error

- Measurement error commonly occurs when using tests or assessments to measure constructs.
- If constructs are not measured perfectly, then the correlation coefficient is attenuated.

Measurement error

- Example: Job performance is not perfectly measured.
  The score (X) is a combination of their true performance (T) and measurement error (ε_1).



Measurement error

```
. generate x_me = x+rnormal(0,.5)

. generate y_me = y+rnormal(0,.5)

. correlate x y x_me y_me
(obs=1,000)
```

	x	y	x_me	y_me
x	1.0000			
y	0.4938	1.0000		
x_me	0.9002	0.4304	1.0000	
y_me	0.4369	0.8985	0.3849	1.0000

Measurement error

- The corrected correlation is given as

$$\rho_{TP} = \frac{\rho_{XY}}{\sqrt{r_{XX} r_{YY}}}$$

where r_{XX} and r_{YY} are their reliability estimates

```
. meta psycor rho n, xreliability() yreliability()
```

Measurement error



Range restriction



Range restriction



Range restriction

- Range restriction is a specific type of sample selection bias in which the variability of scores on one or more variables is artificially reduced because of the selection process.
- Example: You would like to know the correlation between score on a skills test and job performance, but you only have job performance scores from the individuals that were hired.
  - How can we use our restricted (incumbent) sample to make inferences about the unrestricted (applicant) sample?

Range restriction

- Range restriction can be direct or indirect.
  - Direct: The employer only uses the skills test to make hiring decisions. The restriction is directly on X.
  - Indirect: The employer uses the skills test, interview, and past experience to make hiring decisions. The restriction is correlated with X.

Range restriction

- Direct range restriction

```
. correlate x y if x<1
(obs=840)

             |        x        y
-------------+------------------
           x |   1.0000
           y |   0.3745   1.0000
```

- Indirect range restriction

```
. correlate x y if z<0
(obs=484)

             |        x        y
-------------+------------------
           x |   1.0000
           y |   0.4282   1.0000
```

Range restriction

- Range restriction can be direct or indirect.
  - Direct: The employer only uses the skills test to make hiring decisions. The restriction is directly on X.
  - Indirect: The employer uses the skills test, interview, and past experience to make hiring decisions. The restriction is correlated with X.

- Range restriction can be univariate or bivariate.
  - Univariate: All applicants take the skills test. The only restriction is in job performance.
  - Bivariate: Only the applicants with limited past experience took the skills test. There is restriction in both the skills test and job performance.

Range restriction

- Without measurement error, the corrected correlation is given by

$$\rho_{XY,u} = h(u, \rho_{XY,r}) = \frac{\rho_{XY,r}}{u\sqrt{\left(\frac{1}{u^2} - 1\right)\rho_{XY,r}^2 + 1}}$$

- u-ratios are calculated as

$$u = \frac{s_r}{s_u}$$

where s_r is the standard deviation in the restricted sample and s_u is the standard deviation in the unrestricted sample.

```
. meta psycorr rho n, xruratios() yuratios() direct/indirect
```

Range restriction



Range restriction

- With measurement error:
  - Direct range restriction: Correct for range restriction then for measurement error

$$\rho_{TP,u} = \frac{h\left(u_X, \rho_{XP,r}\right)}{\sqrt{r_{xx,u}}}$$

  - Indirect range restriction: Correct for measurement error then for range restriction

$$\rho_{TP,u} = h(u_T, \rho_{TP,r})$$

```
. meta psycorr rho n, xreliability() yreliability()
  xruratios() yuratios() direct/indirect
```

Range restriction



Psychometric meta-analysis

- You can control for all or any combination of
  - Small-study bias
  - Artificial dichotomization
  - Measurement error
  - Range restriction

```
. meta psycorr rho n, small xdich() ydich()
  xreliability() yreliability()
  xuratios() yuratios() direct/indirect
```

Psychometric meta-analysis

- By default, Stata assumes that that your reliability estimates are based on the restricted sample. If they are based on the unrestricted sample,

```
xreliability(rxxu, unrestricted)
```

- By default, Stata assumes that your u-ratios are ratios of the observed data. If they are ratios of true-scores,

```
xruratios(ut, true)
```

Example 1

```
. use https://www.stata.com/exdata/metaslemp.dta
. meta psycorr r N, xreliability(rxx) yreliability(ryy) small
  studylabel(Authors) impute(bootstrap, rseed(12345))
. meta forestplot
. meta galbraithplot
. meta summarize, subgroup(prov)
. meta regress i.prov age
```

Example 2

```
. webuse mapsycorr_uvirr

. meta psycorr rho n, xreliability(rxxr)                     yreliability(ryyr)
xuratios(ux)
. meta psycorr rho n, xreliability(rxxu, unrestricted) yreliability(ryyr)
xuratios(ut, true) indirect
```

References

- Schmidt, F. L., H. Le, and I.-S. Oh. 2019. "Correcting for the distorting effects of study artifacts in meta-analysis and second order meta-analysis". In The Handbook of Research Synthesis and Meta-Analysis, edited by H. Cooper, L. V. Hedges, and J. C. Valentine, 315–338. 3rd ed. New York: Russell Sage Foundation. https://doi.org/10.7758/9781610448864.18.
- Schmidt, F. L., and J. E. Hunter. 2015. Methods of Meta-Analysis: Correcting Error and Bias in Research Findings. 3rd ed. Thousand Oaks, CA: Sage. https://doi.org/10.4135/9781483398105.
- Slemp, G. R., J. G. Field, R. M. Ryan, V. W. Former, A. Van den Broeck, and K. J. Lewis. 2024. Interpersonal supports for basic psychological needs and their relations with motivation, well-being, and performance: A meta-analysis. *Journal of Personality and Social Psychology* 127: 1012-1037. https://doi.org/10.1037/pspi0000459.

Vielen Dank!

Feel free to email me questions at mcain@stata.com